Concepts in Monte Carlo sampling

Gabriele Tartero, Werner Krauth

Mathematical details (Appendix A), Version: 20 Sep 2023

We consider the integral of Eq. (6),

In[ ]:=
```
Integrate[4 / Sqrt[2 (U - x^2/2 - x^4/4)], {x, 0, Sqrt[-1 + Sqrt[1 + 4 U]]}]
```

that Mathematica 13.2 cannot do:

Out[ ]=
$$\int_0^{\sqrt{-1+\sqrt{1+4U}}} \frac{2\sqrt{2}}{\sqrt{U - \frac{x^2}{2} - \frac{x^4}{4}}}\, dx$$

We reconsider the integral for the special value of U = e = exp[1]:

In[ ]:=
```
τ = Integrate[4 / Sqrt[2 (E - x^2/2 - x^4/4)], {x, 0, Sqrt[-1 + Sqrt[1 + 4 E]]}] /. E → U
```

that Mathematica 13.2 succeeds in evaluating:

Out[ ]=
$$4\sqrt{2}\sqrt{\frac{1}{1+\sqrt{1+4U}}}\ \mathrm{EllipticK}\left[\frac{1-\sqrt{1+4U}}{1+\sqrt{1+4U}}\right]$$

Mathematica can also evaluate this integral for integer values of U, and the result agrees with the above. Numerically, we find perfect agreement for all real values of U that we have tested \in [0,20].

In[ ]:=
```
Series[τ, {U, 0, 4}]
```

Out[ ]=
$$2\pi - \frac{3\pi U}{2} + \frac{105\pi U^2}{32} - \frac{1155\pi U^3}{128} + \frac{225225\pi U^4}{8192} + O[U]^5$$

The first two terms in this expression correspond to Eq. (A3) (see also Fig. 3), with the first term being the period of the harmonic oscillator. The expansion for large U is:

In[ ]:=
```
FullSimplify[Series[τ, {U, Infinity, 2}]]
```

Out[ ]=
$$\frac{\sqrt{\pi}\,\mathrm{Gamma}\left[\frac{1}{4}\right]\left(\frac{1}{U}\right)^{1/4}}{\mathrm{Gamma}\left[\frac{3}{4}\right]} - \frac{\sqrt{2}\,\pi^{3/2}\left(\frac{1}{U}\right)^{3/4}}{\mathrm{Gamma}\left[\frac{1}{4}\right]^2} - \frac{\left(\sqrt{\pi}\,\mathrm{Gamma}\left[\frac{5}{4}\right]\right)\left(\frac{1}{U}\right)^{5/4}}{8\,\mathrm{Gamma}\left[\frac{3}{4}\right]} + \frac{\sqrt{\pi}\,\mathrm{Gamma}\left[\frac{7}{4}\right]\left(\frac{1}{U}\right)^{7/4}}{8\,\mathrm{Gamma}\left[\frac{1}{4}\right]} + O\left[\frac{1}{U}\right]^{9/4}$$

Here, the first term corresponds to Eq. (A5) (see also Fig. 3). As indicated in Appendix A, this first term is the period of the quartic oscillator,

In[∘]:=
```
FullSimplify[Integrate[4 / Sqrt[2 (U - x^4/4)], {x, 0, Sqrt[Sqrt[4 U]]}]]
```

Out[∘]=
$$\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{1}{4}\right]}{U^{1/4}\ \mathrm{Gamma}\left[\frac{3}{4}\right]} \quad \text{if} \quad \mathrm{Re}[U] > 0\ \&\&\ U == \mathrm{Re}[U]$$

as indicated in FIg. 3.

The partition function Eq. (13) of the anharmonic oscillator is

In[∘]:=
```
Clear[Z]
```

In[∘]:=
```
Z[β_] = Assuming[β > 0, Integrate[Exp[-β (x^2 / 2 + x^4/4)], {x, -Infinity, Infinity}]]
```

Out[∘]=
$$\frac{e^{\beta/8}\ \mathrm{BesselK}\left[\frac{1}{4}, \frac{\beta}{8}\right]}{\sqrt{2}}$$

A table of values for $\beta = 1,2,\ldots,5$ is:

In[∘]:=
```
Table[N[Z[β], 20], {β, 1, 5}]
```

Out[∘]=
```
{1.9352478184967272764, 1.4863108176097251915,
 1.2622272716816842719, 1.1195578897284502004, 1.0178493404245583598}
```

Finally, the integral of Eq. (C1) is given by

In[∘]:=
```
NIntegrate[Exp[-(x^2 / 2 + x^4/4)], {x, -Infinity, 63/100}, WorkingPrecision → 20] /
 NIntegrate[Exp[-(x^2 / 2 + x^4/4)], {x, -Infinity, Infinity}, WorkingPrecision → 20]
```

Out[∘]=
```
0.80302539727028600 22
```

This result has been confirmed by high-precision numerical integration using the SciPy library.